BUGS code for the network meta-analysis

The model presented here builds on previous work in network meta-analysis (2) and synthesis of data on multiple outcomes (1). The model presented here uses random baselines that correlate the different outcomes across studies, and fixed treatment effects that are estimated separately for each outcome.

model {

  # ND observations. XS, XO, XT index the study, outcome and treatment

  # into the observation vector.

  # Modelled probability p is indexed by study, outcome and treatment.

  for (i in 1:ND) {

    r[i]~dbin(pred[i],n[i])

    pred[i]<-p[XS[i],XO[i],XT[i]]

  }

  # Calculation of trial-specific outcomes, NS studies

  for (i in 1:NS) {

    # Correlated random baseline for the four outcomes

    los[i,1:NO]~dmnorm(mu[],Omega[,])

    for (j in 1:NO) {

      for (k in 1:NT) {

        # Log-odds scale. See Lu and Ades (2004) for matrix A.

        logit(p[i,j,k]) <- los[i,j]+

          inprod(A[,k],tx[((j-1)*(NT-1)+1):(j*(NT-1))])

      }

    }

  }

  # Priors: Vague priors on all the baselines and tx effects

  for (i in 1:NO) {

    mu[i]~dnorm(0,0.001) # Random effects -> los

  }

  for (i in 1:(NO*(NT-1))) {

    tx[i]~dnorm(0,0.0001) # Fixed treatment effects

  }

  # Construct the variance-covariance matrix for the baselines,

  # a priori uncorrelated.

  for (i in 1:NO) { for (j in 1:NO) {

    varcovmu[i,j]<-equals(i,j)

  }}

  # Vague Wishart prior

  Omega[1:NO,1:NO]~dwish(varcovmu[,],NO)

  # Parametrisation of treatment effects, see Lu and Ades (2004).

  # From their matrix A, I moved the placebo treatment to the end.

  for (i in 1:(NT-1)) { for (k in 1:NT) {

    A[i,k] <- equals(i,k) - 1/NT

  }}

}
